

Corrigé Bac S spécialité France juin 2008 à l'aide de XCAS

Guillaume CONNAN
<http://gconnan.free.fr>

23 juin 2008

EXERCICE 1

1. `I:=Int(ln(x),x,1,e)`

$$\int_1^{e^1} \ln(x) dx$$

`J:=Int((ln(x))^2,x,1,e)`

$$\int_1^{e^1} \ln(x)^2 dx$$

(a) `F(x):=x*ln(x)-x`

$$(x) \rightarrow x \cdot \ln(x) - x$$

`deriver(F(x))`

$$\ln(x) + 1 - 1$$

On en déduit I :

`simplifier(F(e)-F(1))`

1

(b) Intégrons par parties J :
$$\begin{cases} u(x) = \ln(x) & u'(x) = \frac{1}{x} \\ v'(x) = \ln(x) & v(x) = F(x) \end{cases}$$

On obtient :

$$J = \left[\ln(x)F(x) \right]_1^e - \int_1^e \ln(x) - 1 dx = 0 - I + (e - 1) = e - 1 - I$$

Or $I = 1$ donc finalement $J = e - 2$.

(c) On en déduit que $J = e - 2$

On vérifie à l'aide de XCAS :

```
int((ln(x))^2,x,1,e)
```

$$e^1 - 2$$

(d) Par lecture du graphique, $A = \int_1^e f(x) - g(x) dx = I - J = 3 - e$

On vérifie à l'aide de XCAS :

```
int(ln(x) - (ln(x))^2,x,1,e)
```

$$-(e^1) + 3$$

2. Les coordonnées de M sont $(x, \ln(x))$ et celles de N $(x, (\ln(x))^2)$ donc

$$MN = \ln(x) - (\ln(x))^2 = f(x) - g(x)$$

Notons $d = f - g$. Étudions la fonction d sur $[1; e]$.

```
d(x) := ln(x) - (ln(x))^2
```

$$(x) \rightarrow \ln(x) - \ln(x)^2$$

Calculons $d'(x)$:

```
dp := factoriser(deriv(d(x)))
```

$$\frac{-(2 \ln(x) + 1)}{x}$$

Étudions son signe :

```
resoudre(dp > 0)
```

$$[(x > 0) \text{ and } (x < (e^{\frac{1}{2}}))]$$

Calculons les valeurs particulières :

```
d(1), d(exp(1/2)), d(e)
```

$$0, \frac{1}{4}, 0$$

On en déduit le tableau suivant :

x	1	$e^{\frac{1}{2}}$	e^1
Signe de $d'(x)$	+	0	-
Variations de d	0	$\frac{1}{4}$	0

Le maximum est donc atteint en $e^{\frac{1}{2}}$ et vaut $\frac{1}{4}$

EXERCICE 2

On vide les mémoires :

```
restart(NULL);;
```

1. (a) `A:=point(1,1,0);B:=point(1,2,1);C:=point(3,-1,2);;`

Done, Done, Done

```
coordonnees(vecteur(A,B))
```

[0,1,1]

```
coordonnees(vecteur(A,C))
```

[2,-2,2]

Les coordonnées des deux vecteurs n'étant pas proportionnelles, on en déduit que les points ne sont pas alignés.

(b) On vérifie que A, B et C appartiennent au plan d'équation $2x + y - z - 3 = 0$.

```
f(C):=2*C[0]+C[1]-C[2]-3;
```

```
f(coordonnees(A)), f(coordonnees(B)), f(coordonnees(C))
```

0,0,0

Les points étant non-alignés, ils définissent le plan.

2. On résout le système $\begin{cases} x + 2y - z - 4 = 0 \\ 2x + 3y - 2z - 5 = 0 \end{cases}$ en prenant z comme paramètre :

```
resoudre([x+2*y-t-4=0,2*x+3*y-2*t-5=0],[x,y])
```

(t-2 3)

3. Comme $(P) \cap (Q) = (D)$, l'intersection des trois plans est donc égale à $(ABC) \cap (D)$.

Un point de (D) ayant pour coordonnées $(-2 + t, 3, t)$, il appartient à (ABC) si, et seulement si, ses coordonnées vérifient l'équation de (ABC) :

```
resoudre(f([-2+t,3,t])=0,t)
```

[4]

On obtient $t = 4$: les trois plans se coupent donc au point de coordonnées $(2, 3, 4)$.

On vérifie avec XCAS :

```
resoudre([x+2*y-z-4=0,2*x+3*y-2*z-5=0,f([x,y,z])=0],[x,y,z])
```

(2 3 4)

4. Soit H le projeté orthogonal de A sur (D). Alors la distance cherchée est égale à AH.
 Comme H appartient à (D), il existe un réel t tel que les coordonnées de H soient $(-2 + t, 3, t)$.
 Un vecteur directeur \vec{u} de (D) a pour coordonnées $(1, 0, 1)$. On a donc $\vec{AH} \cdot \vec{u} = 0$. Le réel t est donc solution de l'équation $(-2 + t) \times 1 + 3 \times 0 + t \times 1 = 0$.

```
H:=point(-2+h,3,h)::
```

```
S:=resoudre(produit_scalaire(vecteur(A,H),vecteur([1,0,1]))=0,h)
```

$$\left[\frac{3}{2} \right]$$

```
H:=point(-2+S[0],3,S[0]);//coordonnees(vecteur(A,H))
```

$$\text{Done,} \left[\frac{-3}{2}, 2, \frac{3}{2} \right]$$

```
distance(A,H)
```

$$\frac{\sqrt{34}}{2}$$

On vérifie :


```
D:=droite([-2+t,3,t],t)::
```

```
distance(A,D)
```

$$\frac{\sqrt{34}}{2}$$

Voici la figure :

```
P:=color(plan(x+2*y-z-4=0),jaune+rempli);
Q:=color(plan(2*x+3*y-2*z-5=0),cyan+rempli);
ABC:=color(plan(2*x+y-z-3=0),magenta+rempli);
D:=inter(P,Q)[0];
I:=inter(D,ABC);
A:=point(1,1,0);
H:=projection(D,A);
segment(A,H);
```



EXERCICE 3

```
restart(NULL) ;;
```

1. (a) $R(t) = 1 - P(X \leq t) = 1 - \int_0^t \lambda e^{-\lambda x} dx = 1 - [-e^{-\lambda x}]_0^t = 1 - (-e^{-\lambda t} + 1) = e^{-\lambda t}$

```
assume (t>0) ;;
```

```
R(t) := simplifier(1 - int(k*exp(-k*x), x, 0, t)) ;;
```

```
R(t)
```

$$e^{-(kt)}$$

(b) $P_{X>t}(X > t + s) = \frac{P((X>t) \cap (X>t+s))}{P(X>t)} = \frac{P(X>t+s)}{P(X>t)} = \frac{R(t+s)}{R(t)} = \frac{e^{-\lambda(t+s)}}{e^{-\lambda t}} = e^{-\lambda(t+s-t)} = R(s)$ qui est indépendant de t .

2. (a) $P(X \leq t) = 1 - R(t)$ donc :

```
k := 0.00026 ;;
```

```
1 - R(1000)
```

0.228948

```
R(1000)
```

0.771052

(b) Comme $P_{X>1000}(X > 1000 + 1000) = R(1000)$, on a $P_{X>1000}(X > 2000) \approx 0,771051585804$

(c) On cherche à calculer

$$P_{X>2000}(X \leq 3000) = 1 - P_{X>2000}(X > 3000) = 1 - P_{X>2000}(X > 2000 + 1000) = 1 - R(1000) = P(X \leq 1000) \approx 0,228948414196$$

EXERCICE 4 Spécialité

```
restart(NULL);
```

1. Sachant que $(1; -1)$ est une solution évidente, on obtient le résultat en utilisant le théorème de Gauss.
2. On se place en mode complexe :

```
complex_variables:=1;; complex_mode:=1;;
```

Done,Done

```
za:=1-i;; zb:=7+7*i/2;; z1:=-2+3*i;;
```

Done,Done,Done

L'angle de la similitude est $(\overrightarrow{AB}, \overrightarrow{AM})$ et le rapport vaut $\frac{AM}{AB}$. Il s'agit donc de calculer le module et l'argument de $Z = \frac{z_1 - z_A}{z_B - z_A}$

```
Z:=(z1-za)/(zb-za)
```

$$\frac{2 * i}{3}$$

```
arg(Z)
```

$$\frac{\pi}{2}$$

```
abs(Z)
```

$$\frac{2}{3}$$

3. Soit f l'application complexe associée à s . Elle est de la forme $az+b$: il s'agit donc d'une similitude directe.

```
f(z):=Z*z+1/3-5*i/3
```

$(z) \rightarrow Z*z+1/3-5*(i)/3$

```
f(za)
```

$$1 - i$$

Le point A est donc invariant par s . Or s n'est pas l'identité donc A est le centre de la similitude directe. Son rapport vaut $|Z| = \frac{2}{3}$ et son angle $\arg(Z) = \frac{\pi}{2} + 2k\pi$, $k \in \mathbb{Z}$.

4. (a) A étant le centre de la similitude, on a $\frac{s(AB_n)}{AB_n} = \frac{2}{3}$ et donc

$$AB_{n+1} = \frac{2}{3}AB_n$$

- (b) Notons (d_n) la suite définie par $d_n = AB_n$. D'après ce qui précède, (d_n) est une suite géométrique de raison $\frac{2}{3}$ et de premier terme

```
abs(zb - za)
```

$$\frac{15}{2}$$

$$d_0 = AB = \frac{15}{2}$$

On a donc

$$d_n = \frac{15}{2} \times \left(\frac{2}{3}\right)^n$$

On cherche donc à résoudre dans \mathbb{N} l'équation $d_n \leq 10^{-2}$

```
no := resoudre((15/2)*(2/3)^n <= 10^(-2), n)
```

$$\left[n \geq \left(\frac{\ln\left(\frac{1}{750}\right)}{\ln\left(\frac{2}{3}\right)} \right) \right]$$

```
evalf(no)
```

$$[n \geq 16.327109]$$

Donc l'entier cherché est 17.

- (c) Les points A, B₁ et B_n sont alignés si, et seulement si, $\left(\overrightarrow{AB_1}, \overrightarrow{AB_n}\right) \equiv 0[\pi]$ c'est-à-dire

$$\arg\left(\frac{z_n - z_A}{z_1 - z_A}\right) = k\pi, \quad k \in \mathbb{Z}$$

Or $z_{n+1} - z_A = \frac{2}{3}iz_n - \frac{2}{3} - \frac{2}{3}i = \frac{2}{3}i(z_n - z_A)$. On en déduit que $z_n - z_A = \left(\frac{2}{3}i\right)^{n-1}(z_1 - z_A)$ et donc que $\arg(z_n - z_A) = (n-1)\frac{\pi}{2} + \arg(z_1 - z_A) + 2k'\pi$, $k' \in \mathbb{Z}$.

On obtient donc $(n-1)\frac{\pi}{2} + 2k'\pi = k\pi$ c'est-à-dire $n = 2(k' - 2k) + 1$ ou encore n est impair.


Vérifions à l'aide de XCAS :

```
s := similitude(za, 2/3, pi/2) ;
```

Pour obtenir l'affixe de l'image de z_B par la composée d'ordre 10 de s , on entre

```
(s@10)(zb) ;
```

```
couleur(legende(za, "A"), bleu+epaisseur_point_3),
droite(point(za), s(zb)), // la droite (AB1),
seq(couleur(legende((s@k)(zb), "B"+k), red+epaisseur_point_2), k=1..9)
```



On remarque que les points de rang impair semblent appartenir à la droite (AB_1) .